
Dafny
- Solver-aided language and verifier 
- Language is statically-typed 
- Imperative (with lots of functional language features) 
- Compiles to C#, Java, Go, Python, …

Reference manual:

https://dafny.org/dafny/DafnyRef/DafnyRef.html



Dafny
- Applies Hoare reasoning to programs 
- User provides specifications in the form of pre- and 
postconditions, along with other assertions 
- Dafny verifies that the program meets the specification 
‣When successful, Dafny guarantees (total) functional correctness of the 
program

3

Correctness:

- Reflects base-level semantic properties (no runtime errors (e.g., 
  divide-by-zero, null pointer dereferences, etc.)
- But, also justifies higher-level application-specific properties (e.g., 
correctness of distributed systems, …)



Types of Program Verification
4

functional 
correctness

limited 
checking

automatic 	
decision	

procedures

interactive	
proof assistants

Rocq

extended static 
checking

Dafny

Adapted from slides of Bryan Parno



Architecture

Pipeline

Adapted from slides of Jon Howell and Manos Kapritsos



Dafny and Separation Logic
‣Verification-aware programming language 

★ Programs are annotated with pre/post-conditions, and invariants (like 
Hoare/SL triples) 
★ A verifier automatically discharges proof obligations via an SMT solver  
★ No separate proof script required 

‣requires clauses are preconditions - { P } 
‣ensures clauses are postconditions - { Q } 
‣invariant clauses express loop invariants 
‣Important differences: 

‣Dafny targets a structured language with objects, arrays, lists, not a raw 
heap of integer-addressed cells 
‣Aliasing and ownership are managed explicitly via framing annotations 
‣Assertion language is first-order over program types, not SL heap 
predicates. 
‣Enforces total correctness

Framing
‣ What does a method modify? 
‣ In SL, the small footprint property and assertions defined via 

separating conjunction answer this question implicitly 
‣ Dafny uses modifies clauses to answer this explicitly 

‣ The modifies clause instructs the verifier that the method may only 
modify a

‣ Any heap location not declared in a modifies clause must remain 
unchanged 

‣ Specifications are bi-state: old(e) refers to the value of the 
expression in the pre-state

method Increment(a: array<int>, i: int) 
  requires 0 <= i < a.Length 
  modifies a 
  ensures a[i] == old(a[i]) + 1



Explicit vs Implicit Reasoning
‣Separation Logic 

• { P * F } c { Q * F } (frame rule) 
•separating conjunction encodes 
disjoint ownership 

•Proofs rely on disjoint heap 
splitting 

•Assertions explicit in each triple 
•Use of recursive predicates (list, 
list_seg) to express higher-level 
heap structure

‣Dafny 
•modifies clause  
•modifies clause sets write footprint 
•collect verification conditions that 
are solved by an SMT solver 

•Declared for each method 
•Support for algebraic datatypes, 
objects, etc.



Specifications
10

- Specifications are meant to capture salient behavior of an application, eliding
  issues of efficiency and low-level representation.

          forall k:int :: 0 <= k < a.Length ==> 0 < a[k]

- Specifications in Dafny can be arbitrarily sophisticated.
  
- We can think of Dafny as being two smaller languages rolled into one:

   - An imperative core that has methods, loops, arrays, if statements... and
     other features found in realistic programming languages.  This core can
     be compiled and executed.

   - A pure (functional) specification language that supports functions, sets,
     predicates, algebraic datatypes, etc.  This language is used by the prover
     but is not compiled.



Examples

method Triple (x: int) returns (r : int) {
  var y := 2 * x;
  r := x + y;
  assert r == 3 * x;
}

method Caller () {
  var t := Triple(9);
  assert t == 27;
}

// assert fails: why?

method TripleSpec (x: int) returns (r : int)
ensures r == 3 * x

{
var y := 2 * x;
r := x + y;

}

method CallerSpec () {
  var t := TripleSpec(9);

assert t == 27; // assert succeeds
}



Examples

// Valid method
method Index (n: int) returns (i : int)

requires 1 <= n
ensures 0 <= i <= n

{
i := n /2;

}

// Invalid assert - how would you fix this?
method CallIndex() {

var t1 := Index(50);
var t2 := Index(50);
assert t1 == t2;

}



Examples

method Min (x : int, y : int) returns (m : int)
  ensures m <= x && m <= y
{ 
   m := if x <= y then x else y;
}

The implementation satisfies the spec but the spec does not capture
the intended behavior!

method Min (x : int, y : int) returns (m : int)
  ensures m <= x && m <= y
  ensures m == x || m == y
{ 
   m := if x <= y then x else y;
}



Modifies

method Swap(a: array<int>, i: int, j: int)
  requires 0 <= i < a.Length && 0 <= j < a.Length
  modifies a
  ensures a[i] == old(a[j]) && a[j] == old(a[i])
  ensures forall k :: 0 <= k < a.Length &&
            k != i && k != j ==> a[k] == old(a[k])

The modifies clause and the forall in the post-condition collectively 
mimic the behavior of the frame rule in SL - the modifies clause 
restricts modifications to a, and the forall it asserts that all cells in a 
other than a[i] and a[j] remain untouched



Functions vs. Methods
- Functions in Dafny have no computational effect 
‣Deterministic 
‣Can be used in specifications! 

function average(a: int, b: int): int {
  (a + b) / 2
}

method Triple (x: int) returns (r: int)
ensures average(r, 3 * x) == 3 * x

{
  if (x < 0) { return -x; } else { return x; }
}

Alternative definition: function average(a: int, b: int): int 
  requires 0 <= a && 0 <= b
{
  (a + b) / 2
}



Functions

function fib(n: nat): nat
{
   if n == 0 then 0 else
   if n == 1 then 1 else
            fib(n - 1) + fib(n - 2)
}

method Fib (n: nat) returns (x: nat)
ensures x == fib(n);

{
  var i := 0;
  x := 0;
  var y := 1;
  while (i < n) {
    x, y := y, x+y;

 i := i + 1;
}

}

Dafny fails to verify this
program.  Why?



Invariants
- Follows the same principle as Hoare logic

method ComputeFib (n: nat) returns (y: nat)
ensures y == fib(n);

{
if (n == 0) { return 0; }

  var i := 1;
  var x := 0;
  y := 1;

while (i < n) 
   invariant 0 < i <= n 

      invariant x == fib (i - 1) 
      invariant y == fib (i)

{
      x, y := y, x+y;

   i := i + 1;
}

}



Invariants

method loopEx (n : nat)
{

var i : int := 0;
while (i < n)

invariant 0 <= i
{ 

i := i + 1;
}
assert i == n;

}

Dafny will not verify this
program.  Why?

Need invariants to be inductive!
  - hold in the initial state
  - hold in every state reachable from the initial state
  - strong enough to imply the postcondition

method loopExCheckFixed (n : nat)
{

var i : int := 0;
while (i < n)

invariant 0 <= i <= n
{ 

i := i + 1;
}
assert i == n;

}



Ghost vs. Compiled

- Ghost constructs are syntactic forms used only in specifications
- Pre- (requires) and post- (ensures) conditions are ghost constructs
- As are assert and invariant

- Some constructs such as functions exist in both ghost and compiled form
- Can explicitly declare variables, parameters, methods, etc. as ghost; such 

objects are not compiled into executables
★Cannot assign ghost entities to complied ones

method Triple(x : int) returns (r: int) 
  ensures r = 3 * x
{
   var y := 2 * x;
   r := x + y;
   ghost var a, b := DoubleQuadruple(x);
   assert a <= r <= b || b <= r <= a;
}

ghost method DoubleQuadruple (x : int) returns (a: int, b: int)
   ensures a = 2 * x && b = 4 * x
{
   a := 2 * x;
   b := 2 * a;
} 



Assert

- assert E is a no-op if E holds, otherwise program faults.
- To show postcondition Q holds, i.e., 
        WP(assert E, Q)
  we must prove E && Q
- Backward reasoning

Alternative interpretation:
- assert E evaluates E and if E does not crash, continues
- No proof obligations introduced
- Forward reasoning



Concept Check

method MultipleReturnsSpec(x: int, y: int) returns (more: int, less: int) 
{
  more := x + y;
  less := x - y;
}

What is a meaningful spec for this method?

method MultipleReturnsSpec(x: int, y: int) returns (more: int, less: int)
  ensures less <= x
  ensures x <= more 
{
  more := x + y;
  less := x - y;
}

The implementation fails to verify
against this spec.  Why?

How would you change the implementation
to conform to the spec?



Concept Check

method Abs(x: int) returns (r: int) 
  ensures r >= 0

{
    if (x < 0)
      { return -x; }
    else
      { return x; }
  }

What’s wrong with this 
spec?  How would you 
fix it?

method AbsFixed(x: int) returns (y: int)
  ensures 0 <= x ==> y == x
  ensures x < 0 ==> y == -x
{
  if (x < 0) { return -x; }
     else { return x; }
} method AbsFixedA(x: int) returns (y: int)

ensures 0 <= y && ( y == x || y == -x)
{
  if (x < 0) { return -x; }
     else { return x; }
}



Concept Check

What precondition do you need to annotate the method with in order  
for the verification to go through? 

method F(x: int) returns (y: int)
// Add a precondition here so that the method verifies.

// Don't change the postconditions.
ensures 0 <= y
ensures 0 <= x ==> x == y
ensures x < 0 ==> y == -x

{
y := x + 2;

}

requires x = -1



Concept Check

method m(n: nat)
{

var i: int := 0;
while i < n

      invariant 0 <= i <= n
{

i := i + 1;
}
assert i == n;

}

// Change this to 0 <= i <= n + 2. What happens?



Concept Check
How would you verify this program?

method FindMax(a: array<int>) returns (max: int)

{
  var i := 1;
  j := 0;

     while index < a.Length

 
 {

 if a[i] >= a[j] { j := i; }
 i := i + 1;

}
}

- what are the pre/post-conditions of the method?
- what are the loop invariants?

requires a.Length > 0
ensures a.Length > 0 ==> 0 <= j < a.Length && 
        forall k :: 0 <= k < a.Length ==> a[k] <= a[j]

invariant 0 <= j < i
invariant 0 < i <= a.Length 
invariant forall m :: 0 <= m < i ==> a[m] <= a[j]



Decreases clause

function seqSum (s : seq<int>, lo : int, hi : int) : int
requires 0 <= lo <= hi <= |s|

{
if (lo == hi) then 0 else s[lo] + seqSum(s, lo+1, hi)

}

Dafny complains that it cannot prove the recursive call terminates - 
it is unable to identify a termination metric that signals every 
recursive call gets “smaller”

function seqSum (s : seq<int>, lo : int, hi : int) : int
requires 0 <= lo <= hi <= |s|

    decreases hi - lo
{

if (lo == hi) then 0 else s[lo] + seqSum(s, lo+1, hi)
}

What about using -lo as a decreases clause?



Examples

function F(x : int) : int {
  if x < 10 then x else F(x - 1)

}

function G(x : int) : int {
if 0 <= x then G(x - 2) else x

}

Are decreases clauses
required?

function H(x : int) : int
 
{

if x < -60 then x else H (x - 1)
}

Are decreases clauses
required here?

decreases x + 60

Why would `decreases x` not work?

function L(x: int) : int
 
{

if x < 100 then L(x + 1) + 10 else x
}

Are decreases clauses
required here?

decreases 100 - x



Well-Founded Relations

A binary relation  is well-founded if it is: 
‣ irreflexive 
‣ transitive 
‣ satisfies a descending chain condition, i.e. there is no infinite sequence of 
values a0, a1, … such that a0  a1  … 

Can establish such a relation for any datatype 
‣ E.g., Booleans (true  false), a less-than ordering relation on integers, or 
a subset relation on a set

⪰

⪰ ⪰

⪰

function M(x: int, b: bool) : int
 
{
  if b then x else M(x + 25, true)
}

decreases if b then 0 else 1



Lexiographic Tuples
- Component-wise comparison of decreases clauses: 

        4, 12  4, 11 

        4, 6, 0  4, 6, 0, 25, 3 

        2, 5  1

⪰
⪰

⪰

function Ack(m : nat, n: nat) : nat
  
{
   if m == 0 then
     n + 1
   else if n == 0 then
     Ack(m - 1, 1)
   else Ack(m - 1, Ack(m, n - 1))
}

What decreases
clause is necessary
to allow Dafny to verify 
this program?

decreases m, n



Lemmas

Sometimes, the property we wish to prove cannot be automatically 
verified.  To help Dafny, we can provide lemmas, theorems that exist 
in service of proving some other property.

method FindZero(a: array<int>) returns (index: int)
  requires forall i :: 0 <= i < a.Length ==> 0 <= a[i]
  requires forall i :: 0 < i < a.Length ==> a[i-1]-1 <= a[i]
{
}

Precondition restricts input array such that all elements are greater than or equal to zero
and each successive element in the array can decrease by at most one from the previous 
element.

We can take advantage of this observation in searching for the first zero in the array, by 
skipping elements.  E.g., if a[j] = 7, then index of next possible zero cannot be before 
a[j + a[j]], i.e., if j = 3, then first possible zero can only be at a[10]



31

Lemmas

method FindZero(a: array<int>) returns (index: int)
  requires forall i :: 0 <= i < a.Length ==> 0 <= a[i]
  requires forall i :: 0 < i < a.Length ==> a[i-1]-1 <= a[i]
  ensures index < 0  ==> forall i :: 0 <= i < a.Length ==> a[i] != 0
  ensures 0 <= index ==> index < a.Length && a[index] == 0
{
  index := 0;
  while index < a.Length
    invariant 0 <= index
    invariant forall k :: 0 <= k < index && k < a.Length ==> a[k] != 0
  {
    if a[index] == 0 { return; }
    index := index + a[index];
  }
  index := -1;
}

Dafny complains about the second loop invariant!

Precondition makes a claim about successive elements,
but invariant relies on generalizing this claim to sequences



Lemmas

lemma SkippingLemma(a: array<int>, j: int)
  requires forall i :: 0 <= i < a.Length ==> 0 <= a[i]
  requires forall i :: 0 < i < a.Length ==> a[i-1]-1 <= a[i]
  requires 0 <= j < a.Length
  ensures forall i :: j <= i < j + a[j] && i < a.Length ==> a[i] != 0
{
  //...
}

Introduce a lemma (a ghost method) that claims all elements from a[j] to a[j + 
a[j]] are non-zero

To use this lemma, “invoke it” in the body of FindZero!
method FindZero(a: array<int>) returns (index: int)
  requires forall i :: 0 <= i < a.Length ==> 0 <= a[i]
  requires forall i :: 0 < i < a.Length ==> a[i-1]-1 <= a[i]
  ensures index < 0  ==> forall i :: 0 <= i < a.Length ==> a[i] != 0
  ensures 0 <= index ==> index < a.Length && a[index] == 0
{
  index := 0;
  while index < a.Length
    invariant 0 <= index
    invariant forall k :: 0 <= k < index && k < a.Length ==> a[k] != 0
  {
    if a[index] == 0 { return; }
    SkippingLemma(a, index);    
   index := index + a[index];
  }
  index := -1;
}



Lemmas

While FindZero now verifies, we still need to provide a proof for SkippingLemma.
First cut: uses asserts to ascertain what Dafny can understand:

lemma SkippingLemma(a: array<int>, j: int)
  requires forall i :: 0 <= i < a.Length ==> 0 <= a[i]
  requires forall i :: 0 < i < a.Length ==> a[i-1]-1 <= a[i]
  requires 0 <= j < a.Length - 3  // strengthened the post-condition

{
  assert a[j  ] - 1 <= a[j+1];
  assert a[j+1] - 1 <= a[j+2];
  assert a[j+2] - 1 <= a[j+3];
  // therefore:
  assert a[j  ] - 3 <= a[j+3];
}

Dafny is able to verify this change of reasoning



Lemmas

lemma SkippingLemma(a: array<int>, j: int)
  requires forall i :: 0 <= i < a.Length ==> 0 <= a[i]
  requires forall i :: 0 < i < a.Length ==> a[i-1]-1 <= a[i]
  requires 0 <= j < a.Length
  ensures forall k :: j <= k < j + a[j] && k < a.Length ==> a[k] != 0
{
  var i := j;
  while i < j + a[j] && i < a.Length
    invariant i < a.Length ==> a[j] - (i-j) <= a[i]
    invariant forall k :: j <= k < i && k < a.Length ==> a[k] != 0
  {
    i := i + 1;
  }
}

The body of the lemma constitutes a proof that its postcondition holds



Lemmas and Induction

Sometimes the property we are interested in proving has a 
natural inductive interpretation

function count(a: seq<bool>): nat
{
  if |a| == 0 then 0 else
  (if a[0] then 1 else 0) + count(a[1..])
}
method m()
{
  assert count([]) == 0;
  assert count([true]) == 1;
  assert count([false]) == 0;
  assert count([true, true]) == 2;
}

Suppose we wish to prove that count distributes over addition…

forall a, b :: count(a + b) == count(a) + count(b)



Lemmas and Induction

lemma DistributiveLemma(a: seq<bool>, b: seq<bool>)
  ensures count(a + b) == count(a) + count(b)
{
}

function count(a: seq<bool>): nat
{
  if |a| == 0 then 0 else
  (if a[0] then 1 else 0) + count(a[1..])
}

- To prove the lemma, we need to structurally decompose the shape(s) of a and b
- We effectively need an inductive proof principle



Lemmas and Induction

lemma DistributiveLemma(a: seq<bool>, b: seq<bool>)
  ensures count(a + b) == count(a) + count(b)
{
  if a == [] {
    assert a + b == b;  
    assert count(a) == 0;
    assert count(a + b) == count(b);
    assert count(a + b) == count(a) + count(b);
  } else {
    //...
  }
}
function count(a: seq<bool>): nat
{
  if |a| == 0 then 0
  else (if a[0] then 1 else 0) + count(a[1..])
}

Base case:



Lemmas and Induction

Inductive case:

lemma DistributiveLemma(a: seq<bool>, b: seq<bool>)
  ensures count(a + b) == count(a) + count(b)

function count(a: seq<bool>): nat
{
  if |a| == 0 then 0
  else (if a[0] then 1 else 0) + count(a[1..])
}

 assert a + b == [a[0]] + (a[1..] + b);
 assert count(a + b) == count([a[0]]) + count(a[1..] + b);



Lemmas and Induction

Express this inductive property:
   assert count(a + b) == count([a[0]]) + count(a[1..] + b);
using recursion

lemma DistributiveLemma(a: seq<bool>, b: seq<bool>)
  ensures count(a + b) == count(a) + count(b)
{
  if a == [] {
    assert a + b == b;
  } else {
    DistributiveLemma(a[1..], b);
    assert a + b == [a[0]] + (a[1..] + b);
  }
}
function count(a: seq<bool>): nat
{
  if |a| == 0 then 0 else
    (if a[0] then 1 else 0) + count(a[1..])
}



Proof Calculations

lemma prop_app_Nil(xs: list)
  ensures app(xs, Nil) == xs;
{

match xs {
case Nil =>
case Cons(y,ys) =>

calc { app(xs,Nil) ;
== app(Cons(y,ys), Nil);
== Cons(y, app(ys,Nil));
== { prop_app_Nil(app(ys,Nil)); }  // proof hint

           xs;
           }

}
}

Proof that Nil is idempotent over list appends



Proof Calculations

Adapted from slides of Bryan Parno

Constructive proofs that involve rewrites and simplification

calc {
  (x + y) * (x - y);
==
  (x * x) - (x * y) + (y * x) - (y * y);
== 
  (x * x) - (x * y) + (x * y) - (y * y);
==
  (x * x) - (y * y);
}



Proof Calculations

Adapted from slides of Jon Howell and Manos Kapritsos

Proof that list append is associative

lemma prop_app_assoc(xs: list, ys: list, zs: list)
  ensures app(xs, app(ys, zs)) == app(app(xs, ys), zs);
{

match xs {
case Nil =>

calc { app(Nil, app(ys, zs));
== app(app(Nil, ys), zs);
}

case Cons(hd,tl) =>
            calc { app(Cons(hd,tl), app(ys, zs));

== Cons(hd, app(tl, app(ys, zs)));
== { prop_app_assoc(tl, ys, zs); }
Cons(hd, app(app(tl, ys), zs));
== app(app(Cons(hd,tl), ys), zs);
== app(app(xs, ys), zs);
}

}
}



Proof Calculations and Induction

datatype Tree<T>  =
  Leaf(data : T)

| Node(left: Tree<T>, right : Tree<T>)

function mirror<T>(t : Tree<T>) : Tree<T> {
match t

case Leaf(_) => t
case Node(left, right) => Node(mirror(right), mirror(left))

}

lemma {:induction false} MirrorMirror<T>(t: Tree)
  ensures mirror(mirror(t)) = t
{
   // Proof that mirror is involutive
}



Proof Calculations and Induction

lemma {:induction false} MirrorMirror<T>(t: Tree)
ensures mirror(mirror(t)) == t
{

match t
case Leaf(_) =>
case Node(left,right) =>

calc
{

mirror(mirror(Node(left,right)));
==
mirror(Node(mirror(right),mirror(left)));
==
Node(mirror(mirror(left)),mirror(mirror(right)));
==  // IH

                 { MirrorMirror(left); MirrorMirror(right); }
                Node(left, right);

}
}



Proof Calculations and Induction

function size<T>(t : Tree<T>): nat {
  match t

case Leaf(_) => 1
case Node(left, right) => size(left) + size(right)

}

lemma {:induction false} MirrorSize<T>(t : Tree<T>)
ensures size(mirror(t)) == size(t)

{
 

match t
    case Leaf(_) =>

      case Node(left, right) =>
calc {
size(mirror(Node(left,right)));

==
  size(Node(mirror(right),mirror(left)));

==
size(mirror(right)) + size(mirror(left));

==
{ MirrorSize(right); MirrorSize(left); } // I.H
size(right) + size(left);

==
             size(Node(left,right));

}

}



Proofs by Contradiction
General shape: 

lemma Lem(args)

   requires P(x)

   ensures Q(x)

 { 

    if !Q(x)                // property is false

      {

        assert !P(x)        // contradiction: precondition is 

        assert false        // true and false

    }

   assert Q(x)

}

46

!Q -> (R /\ !R)
———————————————-
      Q

slide adapted from Albert Nymeyer



Proof by Contradiction

lemma L(i : int)
  requires i > 42
  ensures i > 0
 {
   if i <= 0 {
     assert i <= 0 && i > 42;   // define contradiction
     assert false;
   }
  }

Dafny can verify this claim without the need of a lemma in this case, 
but in general the property may require reasoning beyond Dafny’s 
ability to discharge automatically

slide adapted from Albert Nymeyer



Proof by Contradiction
Claim: if s is a singleton set, and i \in s, then s = { i }

48

lemma isSingle(s: set<int>, i: int)
  requires |s| == 1 && i in s
  ensures s == {i}
 {
   if s > {i}    // {i} is a subset of s
     {
        assert |s - {i}| >= 1 ==> |s| > 1;
        assert |s| == 1 && |s| > 1; // from precondition and branch
        assert false;
     }
 }

slide adapted from Albert Nymeyer



Proof by Contradiction
Given two disjoint sets A and B, if x \in A then x !\in B

lemma Disjoint(a: set<int>, b: set<int>, x: int)
  requires a * b == {}  // a and b are disjoint
  requires x in a
  ensures x !in b
 {
   if x in b      // setup contradiction
    {
      assert (x in b) && (x in a);  // from branch and pre-condition
      assert a*b == {x};  // from assert and precondition
      assert a*b == {x} && a*b == {}; // from assert and precondition
      assert false;
    }
}

slide adapted from Albert Nymeyer



Proof by Contradiction
Usage:

50

method Validate() {
  var a: set<int>, b: set<int>;
  var x: int;

  if x in a && a*b == {} {
    Disjoint(a, b, x);  // eliminating this proof causes the
    assert(x !in b);    // assert to fail
  }
}

slide adapted from Albert Nymeyer



Assume

method ExFalsoQuodlibet() {
  var f := true;
  assume !f   // let f be logically false, so f /\ !f
  var pigscanfly := false;
  assert pigscanfly;
}

slide adapted from Albert Nymeyer

Dafny will verify (but not compile) this program.  Why?

method IdVerum() {
  var f := true;
  assume f;
  var pigscanfly := false;
  assert pigscanfly;
}

Dafny will not verify this program

method EsseVerum() {
  var f : bool;
  assume f;
  var pigscanfly := false;
  assert pigscanfly;
}

Dafny will not verify this program


